9 数学関数

Mathematica に組み込まれている種々の数学関数について解説する．この章で解説する関数はほとんどすべてが Listable 属性を持っているので，それらの引き数としてリストを与えると，そのリストの各要素に関数を施したリストを返す．

■ 数値操作関数

数値の整数値への変換や，数値の符号，絶対値，最大値，最小値などを求める関数である．

■ 数値の整数への変換

整数でない数値を，それに近い整数に変換する関数としては，整数部分をとる IntegerPart，最も近い整数にする Round，切り下げる Floor，切り上げる Ceiling などがある．また整数部分を取り除いた残りである小数部分をとる関数 FractionalPart もここで説明しておく．ちなみに， integer は整数，round は丸める， floor は床， ceiling は天井， Fractional は端数の，という意味である．

これらの関数は，実数だけではなく，複素数に対しても働く．その場合，実数部分，虚数部分それぞれに働く．

● IntegerPart[x] は x の整数部分を返す．

```
In[41]:= IntegerPart[3.7]
Out[41]= 3

In[42]:= IntegerPart[3.7 - 2.1 I]
Out[42]= 3 - 2 i

In[43]:= IntegerPart[1000000 Pi]
Out[43]= 3141592
```

● FractionalPart[x] は x の小数部分を返す．

```
In[44]:= FractionalPart[{3.7, 3.5 - 2.1 I, Pi}]
Out[44]= {0.7, 0.5 - 0.1 i, -3 + π}
```

● Round[x] は x に最も近い整数（整数への丸め）を返す．

```
In[45]:= Round[{3.4, 3.7, 3.7 - 2.1 I}]
Out[45]= {3, 4, 4 - 2 i}
```

小数部分がちょうど 0.5 であるときには，下に丸めるべきか上に丸めるべきかがはっきりしない．そこで， Round[n+0.5] は n が偶数のときには下に，奇数のときには上に丸めることにしている．

x	-4.5	-3.5	-2.5	-1.5	-0.5	0.5	1.5	2.5	3.5	4.5
Round[x]	-4	-4	-2	-2	0	0	2	2	4	4

● Floor[x] は x 以下の最大の整数を返す.

```
In[46]:= Floor[{3.7, -3.7}]
Out[46]= {3, -4}
```

● Ceiling[x] は x 以上の最小の整数を返す.

```
In[47]:= Ceiling[{3.7, -3.7}]
Out[47]= {4, -3}
```

■ 符号と絶対値

数値の符号は Sign, 絶対値は Abs という関数を用いる. Abs は absolute value （絶対値）の略である.

● Sign[x] は, 実数 x が正であるとき +1 を, 0 であるとき 0 を, 負であるとき -1 を返す.

```
In[48]:= Sign[{-3, 0, 3}]
Out[48]= {-1, 0, 1}
```

● Abs[x] は実数または複素数の絶対値を返す.

```
In[49]:= Abs[{-3, 1 + 2 I}]
Out[49]= {3, Sqrt[5]}
```

■ 最大値と最小値

● Max[x1, x2, ...] あるいは Max[{x1, x2, ...}, ...] は x1, x2, ... の中の最大値を返す.

```
In[50]:= Max[3, 2, 1, 5, 2, 3, 4]
Out[50]= 5

In[51]:= Max[{3, 2, 1}, {5, 2, 3, 4}]
Out[51]= 5
```

● Min[x1, x2, ...] あるいは Min[{x1, x2, ...}, ...] は x1, x2, ... の中の最小値を返す.

```
In[52]:= Min[3, 2, 1, 5, 2, 3, 4]
Out[52]= 1

In[53]:= Min[{3, 2, 1}, {5, 2, 3, 4}]
Out[53]= 1
```

■ 複素数に関する関数

● Re[z] は複素数 z の実数部分を返す.

```
In[54]:= Re[2 + 3 I]

Out[54]= 2
```

● Im[z] は複素数 z の虚数部分を返す.

```
In[55]:= Re[2 + 3 I]

Out[55]= 2
```

● Conjugate[z] は複素数 z の共役複素数を返す.

```
In[56]:= Conjugate[2 + 3 I]

Out[56]= 2 - 3 i
```

● Abs[z] は複素数 z の絶対値を返す.

複素数 $a+b\,i$ の絶対値は $\sqrt{a^2+b^2}$ である.

```
In[57]:= Abs[2 + 3 I]

Out[57]= √13
```

● Arg[z] は複素数 z の偏角を返す.

```
In[58]:= Arg[2 + 2 I]

Out[58]= π/4
```

■ 乱数（疑似乱数）

乱数（正確には疑似乱数）は Random という関数を用いて生成することができる．また， SeedRandom を用いると，乱数の初期値（種）を指定することができる.

■ Random

● Random[Integer] は 0 と 1 （各々 $\frac{1}{2}$ の確率）を値域とする整数値乱数を返す.

● Random[Integer, n] は 0, 1, 2, ..., n （各々 $\frac{1}{n+1}$ の確率）を値域とする整数値乱数を返す.

● Random[Integer, {min, max}] は min, ..., max （各々 $\frac{1}{\text{max}-\text{min}+1}$ の確率）を値域とする整数値乱数を返す.

```
In[59]:= Table[Random[Integer, 2], {10}]

Out[59]= {2, 0, 2, 2, 1, 0, 0, 2, 1, 1}
```

- Random[Real] あるいは Random[] は， 0 から 1 の間の実数値乱数を返す.
- Random[Real, max] は 0 から max の間の実数値乱数を返す.
- Random[Real, {min, max}] は min から max の間の実数値乱数を返す.

```
In[60]:= Table[Random[], {4}]

Out[60]= {0.306057, 0.83066, 0.551223, 0.850865}
```

- Random[Complex] は，実数部分，虚数部分共に 0 から 1 の間にあるような複素数値乱数を返す.
- Random[Complex, {zmin, zmax}] は，実部，虚部がそれぞれ zmin と zmax の実部，虚部の間にあるような複素数値乱数をかえす.

```
In[61]:= Table[Random[Complex, {-1 - I, 1 + I}], {3}]

Out[61]= {0.534871 + 0.0955265 i, 0.544335 + 0.456184 i, -0.518704 - 0.0943282 i}
```

- Random[type, range, n] は type の型で， range の範囲内の， n 桁の精度を持つ乱数を返す.

```
In[62]:= Random[Real, {0, 1}, 20]

Out[62]= 0.066370811721762624572
```

■ SeedRandom

Random は一つの種 (seed) をもとに疑似乱数列を生成する． SeedRandom はその種を設定するための関数である.

- SeedRandom[] は現在時間を元にした値を Random の種に設定する.

現在時間を元にして乱数の種を設定すると， Random の返す乱数列が同じものとなることは，ほとんどなくなる． *Mathematica* の起動時には SeedRandom[] が呼び出されるので， *Mathematica* を起動する度に， Random は異なった乱数列を返すことになる.

- SeedRandom[s] は s を Random の種に設定する.

同じ値を種に設定すると， Random が返す乱数列はいつでも同じものとなる.

```
In[63]:= SeedRandom[5]

In[64]:= Table[Random[], {3}]

Out[64]= {0.786599, 0.848339, 0.612827}

In[65]:= SeedRandom[5]

In[66]:= Table[Random[], {3}]

Out[66]= {0.786599, 0.848339, 0.612827}
```

練習問題： 1 以上 10000 以下の整数の乱数を 100 個作ってリストにせよ.

■ 整数論に関連した関数

ここでは，整数の位取り表記，商，剰余，最大公約数，素因数分解などを求める関数，素数に関する関数，そして組合せ関数について説明する．

■ 位取り表記

整数を位取り表記法で表したときの各桁の数字のリストは IntegerDigits を用いて得られる．

● IntegerDigits[n, b] は，整数 n を b 進数で表したときの各桁の数字のリストを返す． b を省略すると 10 進数を指定したことを意味する．

```
In[67]:= IntegerDigits[10, 2]

Out[67]= {1, 0, 1, 0}
```

IntegerDigits[n, b, len] は， IntegerDigits[n, b] で求めたリストの先頭に幾つかの 0 を付け加えて長さを len にしたものを返す．

```
In[68]:= IntegerDigits[10, 2, 8]

Out[68]= {0, 0, 0, 0, 1, 0, 1, 0}
```

練習問題： 貴方の学生番号の各桁の数字のリストを IntegerDigits を用いて作れ．また，それに Plus を適用(Apply)することにより，それらの数字の総和を求めよ．

■ 商と剰余

整数 a を整数 b で割ったときの商 (quotient) q と余り (remainder) r は $a = b\,q + r$ $(0 \le r < b)$ を満たす整数として定義される．また r は b を法 (modulo) とした a の値とも言う．

● Quotient[a, b] は a を b で割ったときの商を返す．

```
In[69]:= Quotient[10, 3]

Out[69]= 3
```

● Mod[a, b] は a を b で割ったときの余りを返す．

```
In[70]:= Mod[10, 3]

Out[70]= 1
```

a を 3 で割ったときの商と余りは次のように変化する．特に a が負のときの値に注意．

a	-6	-5	-4	-3	-2	-1	0	1	2	3	4	5	6
Quotient[a, 3]	-2	-2	-2	-1	-1	-1	0	0	0	1	1	1	2
Mod[a, 3]	0	1	2	0	1	2	0	1	2	0	1	2	0

Quotient, Mod は実数に対しても働く.

```
In[71]:= Quotient[10.2, 3.1]
Out[71]= 3
In[72]:= Mod[10.2, 3.1]
Out[72]= 0.9
```

この場合でも, $a = b\,q + r$ は成立している.

```
In[73]:= 3.1 * 3 + 0.9
Out[73]= 10.2
```

- PowerMod[a, b, n] は, Mod[a^b, n] すなわち a^b を n で割った余りを返す.

```
In[74]:= PowerMod[11, 100000000000000000000, 999]
Out[74]= 988
```

PowerMod[a, b, n] は Mod[a^b, n] よりも格段に早く答えを返す. 上の計算結果を Mod[11^100000000000000000000, 999] によって求めるのは, 実際には不可能である.

■ 最大公約数と最小公倍数

- GCD[a, b, ...] は a, b, ... の最大公約数 (greatest common divisor) を返す.

```
In[75]:= GCD[123, 321]
Out[75]= 3
```

- GCD[a, b, ...] は a, b, ... の最小公倍数 (least common multiple) を返す.

```
In[76]:= LCM[123, 321]
Out[76]= 13161
```

■ 素因数分解と約数のリスト

整数の素因数分解には FactorInteger を, 約数のリストを得るには Divisors を用いる.

- FactorInteger[n] は整数 n の素因数とその指数のリストを返す.

```
In[77]:= FactorInteger[360]
Out[77]= {{2, 3}, {3, 2}, {5, 1}}
```

この結果は $360 = 2^3\ 3^2\ 5^1$ を意味している.

- Divisors[n] は整数 n の約数のリストを返す.

```
In[78]:= Divisors[36]

Out[78]= {1, 2, 3, 4, 6, 9, 12, 18, 36}
```

練習問題： 貴方の学生番号を素因数分解せよ．またその約数のリストを求めよ．

■ 素数

素数とは 1 とそれ自身以外の約数を持たないような整数 2, 3, 5, 7, 11, ... のことである．素数に関しては Prime, PrimePi, PrimeQ などの関数がある．

● Prime[n] は n 番目の素数を返す．

```
In[79]:= Table[Prime[n], {n, 10}]

Out[79]= {2, 3, 5, 7, 11, 13, 17, 19, 23, 29}

In[80]:= Prime[10000]

Out[80]= 104729
```

Prime[n] は n が 10^9 くらいまでは瞬時にその値を返す．

● PrimePi[x] は x 以下の素数が何個あるかを返す．

```
In[81]:= PrimePi[104729]

Out[81]= 10000
```

● PrimeQ[n] は n が素数であるか否かを返す．

```
In[82]:= PrimeQ[104729]

Out[82]= True
```

PrimeQ[n] が返す値は， n < 2.5 10^{10} の範囲では正しい．それを超えるような大きい n に関しては，素数でない n に対して True を返す可能性があるが，その確率は非常に小さい．

練習問題： 1 以上 10000 以下の整数の乱数を 100 個作ってリストにせよ．その中から素数であるものを選びだし，リストにせよ．

■ 組合せ関数

組合せ論に関する関数も種々あるが，基本的なものには Factorial (!), Binomial, Multinomial などがある．

● Factorial[n] または n! は， n の階乗 1×2×⋯×n を返す．

これは n 個のものの並べ方の総数である．

```
In[83]:= 10!
Out[83]= 3628800
```

● Binomial[m, n] は２項係数 ${}_mC_n = \binom{m}{n} = \frac{m!}{n!\,(m-n)!}$ を返す.

これは m 個のものから n 個のものを選ぶ方法の総数である.

```
In[84]:= Binomial[100, 10]
Out[84]= 17310309456440
```

● Multinomial[n1, n2, ...] は多項係数 $\frac{(n_1+n_2+\cdots)!}{n_1!\,n_2!\cdots}$ を返す.

これは $n_1 + n_2 + \cdots$ 個のものを n_1 個， n_2 個，... に分ける方法の総数である.

```
In[85]:= Multinomial[3, 4, 5]
Out[85]= 27720
```

練習問題： 100 個のものを 20 個と 30 個と 50 個とに分ける方法は何通りあるか.

■ 初等超越関数

指数関数，対数関数，三角関数，逆三角関数などを総称して，初等超越関数と呼ぶ.

■ 指数関数，対数関数

Exp[x] は指数関数 e^x である.

Log[x] は自然対数関数 $\log_e x$ である.

Log[b, x] は b を底とする対数関数 $\log_b x$ である.

■ 三角関数

Mathematica の三角関数の引き数の単位はラジアンである.



Sin[x] は正弦関数 $\sin(x) = \frac{a}{c}$ である.

Cos[x] は余弦関数 $\cos(x) = \frac{b}{c}$ である.

Tan[x] は正接関数 $\tan(x) = \frac{a}{b}$ である.

Sec[x] は正割関数 $\sec(x) = \frac{1}{\cos(x)} = \frac{c}{b}$ である.

Csc[x] は余割関数 $\mathrm{cosec}(x) = \frac{1}{\sin(x)} = \frac{c}{a}$ である.

Cot[x] は余接関数 $\cot(x) = \frac{1}{\tan(x)} = \frac{b}{a}$ である.

```
In[86]:= Tan[Pi / 3]
```

Out[86]= $\sqrt{3}$

単位を"度"にする場合には，引き数に Degree をつければよい.

```
In[87]:= Tan[60 Degree]
```

Out[87]= $\sqrt{3}$

■ 逆三角関数

逆三角関数とは，三角関数の逆関数のことである．関数値の単位はラジアンである.

ArcSin[x] は逆正弦関数 $\arcsin(x)$ である．値域は $[-\frac{\pi}{2}, \frac{\pi}{2}]$ である.

ArcCos[x] は逆余弦関数 $\arccos(x)$ である．値域は $[0, \pi]$ である.

ArcTan[x] は逆正接関数 $\arcsin(x)$ である．値域は $[-\frac{\pi}{2}, \frac{\pi}{2}]$ である.







ArcCsc[x], ArcSec[x], ArcCot[x] はそれぞれ $\mathrm{cosec}(x)$, $\sec(x)$, $\cot(x)$ の逆関数である.

```
In[88]:= ArcTan[Sqrt[3]]
```

Out[88]= $\frac{\pi}{3}$

ArcTan[x, y] は複素数 $x + y\,i$ の偏角を返す． $x > 0$ のときは ArcTan[y/x] と同じである.

```
In[89]:= ArcTan[1, Sqrt[3]]
```

Out[89]= $\frac{\pi}{3}$

■ 演習問題

[9-1] 円周率の最初の 100 桁の数字のリストを， Floor[10^99 Pi] に IntegerDigits を施すことにより作れ． 0, 1, 2, ..., 9 の数字の中で，円周率の最初の 100 桁に一番多く出てくるのはどれか．

[9-2] Round[n+0.5] は n が偶数のときには切り下げて，奇数のときには切り上げる．このようにすることの利点を一つあげよ．

[9-3] Table[{Random[], Random[]}, {100}] を実行すると，乱数でできた {x, y} の形のリストのリストができる．この中から $x^2 + y^2 < 1$ であるようなものを選び出し，その個数を数て，全体の個数 100 との比を求めよ．同様の事を 1000 個のリストに対して行い，同様の比を求めよ．そして，その比が円周率の４分の１ ($\frac{\pi}{4}$) に近いことを確認せよ．

[9-4] 貴方の学生番号を a とし，その数字を逆順にした数を b とする． a と b の最大公約数および最小公倍数を求めよ．

[9-5] 100 人の中から 10 人を選ぶ場合の数と， 16 人が一列に並ぶ場合の数はどちらが多いか．

[9-6] 10000 から 20000 の間には素数が何個あるか．